
186 

 “ . ”,  53, .I , , , 2010 
ANNUAL of the University of Mining and Geology “St. Ivan Rilski”, Vol. 53, Part , Mechanization, electrification and automation in mines, 2010 

 
 
 
 
 
 

 
 

 
 

 
 " . ", 1700 , -mail: nre@abv.bg 

 
. ,  

 Intel Pentium  SSE  30%  
. 

 
ARCHITECTURE DEPENDANT REALIZATION OF FAST FOURIER TRANSFORM BY USING SSE INSTRUCTION SET 
Yasen Vidolov 
University of Mining and Geology “St. Ivan Rilski”, 1700 Sofia, e-mail: nre@abv.bg 
 
ABSTRACT. The suggested Fast Fourier Transform code utilizes Intel's Pentium SSE instruction set. The code written in assembler is highly dependent on the 
architecture. It reduces computation time with 30% than its C-code equivalent. 

 mov edx, [n] 

shl edx, 1  //n = 2*n  
 mov [n2], edx 
 shl edx, 1  //n = 4*n because var size is 4B 
 add edx, vec  //n_vec = n + (int)vec;  
 mov [n_vec], edx 
 mov esi,  8  //mov esi, [mmax] //mmax is esi 
 mov edi,  8  //mov edi, [mmax] //istep is edi 
 

loop_A:     //------------------------------------------ 
 shl edi,  1  //istep = 2 * mmax 
     //mmax_vec = mmax + (int)vec; 
 mov edx, esi   //edx is temporal reg 
 add  edx, vec 
 mov [mmax_vec],edx 
 
 // theta = -8.0f * pi / (float)mmax; 
 // wtemp = sin(0.5*theta); 
 // wp[0] = -2.0 * wtemp * wtemp;  //wpr = cos(theta) - 1 
 // wp[1] = sin(theta); //wpi 
 movss xmm0, [pi8]  //xmm1 = -8*pi 
 CVTSI2SS xmm1, esi  //convert mmax to float and save it in xmm1 
 divss xmm0, xmm1  //xmm0 = -8*pi/max (theta)   !!! possible division by 0!!!Add exception here 
 pshufd xmm0, xmm0, 0h  //xmm0 (theta, theta, theta, theta) copy theta to all 4 floats 
 movlps xmm2, [mul_theta]  //xmm2 (0.5, 1.0), x, x 

mulps xmm0, xmm2  //xmm0 (0.5*theta, theta), x, x 
 call sin_ps   //4 float sin functions (in rad): xmm0 = sin(xmm0) 
                        //xmm0 (sin(0.5*theta), sin(theta), x, x) 
 movlps xmm1, [mul_theta]  //xmm1 (0.5, 1, x, x) 
 movss xmm1, xmm0  //xmm1 (sin(0.5*theta), 1, x, x) 
 mulps xmm0, xmm1  //xmm0 (sin(0.5*theta)^2, sin(theta), x, x) 

mulss xmm0, [two]  //xmm0 (-2*sin(0.5*theta)^2, sin(theta), x, x) 
 movlps [wp], xmm0  //move xmm0 to (wp[0], wp[1]) 
 
 //initialize (wr, wi) with (1, 0) 
 //xmm6 contains current (wr, wi) values 
 movlps xmm6, [cw]   //xmm6 (1,0,x,x) 
 mov eax, [vec]   //counter 'm' is eax 
loop_B: mov ebx, eax   //counter 'i' is ebx 
  mov ecx, ebx  //ecx is i2 
  sub ecx, edi 
loop_C:  add ecx, edi  //add istep  
   add ecx, edi //add istep  
 
   //create vector (wr,-wi, wr, wi)  
   movaps xmm0, xmm6 //xmm0 (wr, wi, x, x) 
   pshufd xmm0, xmm0,44h //xmm0 (wr, wi, wr, wi) 44h = Lo %01 00 01 00 Hi (order: 00 01 10 11) 
   mulps xmm0, [signs] //xmm0 (wr,-wi, wr, wi) 
     
   //start butterfly calculation 
   movlps xmm1, [ebx] //xmm1 (v1[i], c1[i], x, x) 
   movlps xmm2, [ebx + esi] //xmm2 (v1[j], c1[j], x, x) 
 
   movhps xmm1, [ecx] //xmm1 (v1[i], c1[i],     v2[i], c2[i]) 
   movhps xmm2, [ecx + esi] //xmm2 (v1[j], c1[j],     v2[j], c2[j]) 
 
   pshufd xmm3, xmm2,14h //xmm3 (v1[j], c1[j], c1[j], v1[j]) 14h = %00 01 01 00 
   pshufd xmm4, xmm2,0BEh //xmm4 (v2[j], c2[j], c2[j], v2[j]) 14h = %10 11 11 10 
 
   //calculate (tempr, tempi) 
   mulps xmm3, xmm0 //xmm3 (v1[j]*wr, c1[j]*wi, c1[j]*wr, v1[j]*wi) 
   mulps xmm4, xmm0 //xmm4 (v2[j]*wr, c2[j]*wi, c2[j]*wr, v2[j]*wi) 
   haddps xmm3, xmm4 //xmm3(v1[j]*wr+c1[j]*wi, c1[j]*wr+v1[j]*wi, v2[j]*wr+c2[j]*wi,  
      //  c2[j]*wr+v2[j]*wi) , (tempr1, tempi1, tempr2, tempi2) 
   //here xxm4 is free     
   //calculate new values of v[i], c[i] 
   //v[i] = v[i] + tempr; 
   //c[i] = c[i] + tempi; 
   movaps xmm2, xmm1 //xmm2 (v1[i], c1[i],     v2[i], c2[i]) 
   addps xmm2, xmm3 //xmm2 (v1[i]+tempr1, c1[i]+tempi1, v2[i]+tempr2, c2[i]+tempi2) 
 
   movlps [ebx], xmm2 //store new calculated values of (v1[i], c1[i]) 
   movhps [ecx], xmm2 //store new calculated values of (v2[i], c2[i]) 
 
   //calculate new values of v[j], c[j] 
   //v[j] = v[i] - tempr; 
   //C[j] = C[i] - tempi; 
   subps xmm1, xmm3 //xmm1(v1[i]-v1[j]*wr+c1[j]*wi, c1[i]-c1[j]*wr+v1[j]*wi,  
      //  v2[i]-v2[j]*wr+c2[j]*wi, c2[i]-c2[j]*wr+v2[j]*wi) 
   movlps [ebx + esi], xmm1 //store new calculated values of (v1[j], c1[j]) 
   movhps [ecx + esi], xmm1 //store new calculated values of (v2[j], c2[j]) 
 
  //loop_C end condition 
  add ebx, edi  //step 
  add ebx, edi  //step2 
  cmp ebx, [n_vec]  //end condition 
  jb loop_C   //Jump If less 
 
  //calculate new normal vector for CORDIC algorithm 
  //wr = (wtemp = wr) * wpr - wi * wpi + wr; 
  //wi = wi * wpr + wtemp * wpi + wi; 
   
  //xmm0 is (wr,-wi, wr, wi) 
  movlps xmm4, [wp]  //xmm4 (wpr, wpi, x, x) 
  pshufd xmm4, xmm4, 14h //xmm4 (wpr, wpi, wpi, wpr) 14h = Lo %00 01 01 00 Hi (order: 00 01 10 11) 
  mulps xmm4, xmm0 
  haddps xmm4, xmm4 
  addps xmm6, xmm4 
 
 //loop_B end condition 
 add eax, 8   //step = 2*4B 
 cmp eax, [mmax_vec]   //end condition 
 jb loop_B    //Jump If less 
 //mmax = istep 
 mov esi, edi   //mmax is esi //or shl esi, 1 
 cmp esi, [n2]   //if(mmax < n/2) loop again 
 jb loop_A  

8 1508 1219 19.2 
16 2436 1786 26.6 
32 4460 3188 28.5 
64 8539 6115 28.4 

128 16320 11712 28.2 
256 32778 22702 30.7 
512 68232 46754 31.5 
1024 144331 96465 33.2 
2048 308185 202734 34.2 
4096 664560 430148 35.3 

